NewBear Computing Store
A Division of NEWBURY LABORATORIES LTD.

SALES & SERVICE: 7 BONE LANE NEWBURY BERKSHIRE RG14 5SH NEWBURY (0635) 49223

DESIGN NOTE 35

6800 PROM PROGRAMMER

£ .60

Instructions

The 2708 PROM supplied with the unit contains the software required to run the PROM programmer program in a 6800 based system.

The PROM can be addressed anywhere in memory as it has been written using position independent code. This means that the PROM can be installed in a system irrespectively of the memory address at which it is situated, and therefore can be memory mapped in any convenient part of memory on a particular system. The program itself starts from the first location of the PROM. That is, if the PROM is addressed at location C000 hex, the program can be started from the same location.

The PROM programmer used MIKBUG/SWTBUG/SMARTBUG/RT68 routines for input-output, these are explained in the listing.

The PROM program uses a single PIA located at address 8100 hex.

Setting up

1. Connect the PROM programmer unit to the microprocessor via a PIA devioe addresses at location 8100 hex. If your system has a PIA at a different address, do not worry as this program can be patched as it will be shown in the next section.

2. Before plugging in the PROM or the power supply connections, ensure the program switch is in the off position, this will stop the 26 volts been applied to the program pin of the chip, better still, use a pushbutton for the program switch.

3. If the system supports a 2708 socket, put the PROM in this socket and run the program from the first address at which the socket has been mapped into the system. If your PIA resides at location other than 8100 hex, or you wish to make changes to the program, copy the program from the PROM to any location in user ram and perform any changes required, save this program as it can be used to program it into a new PROM configured for your system.

4. If your system does not have a socket for a 2708, use the enclosed program to copy the contents of the PROM, which is now put in the socket of the programmer into RAM, as with 3 above, perform any changes required to the program.

Operation

Run the program from the first location, also enter the first and last address of ram to be copied into PROM at addresses 0000/1 and 0002/3. According to 2708 data sheets, program reliability is improved when all locations are programmed on each run, so enter all 1024 locations at the two addresses i.e. if 1024 bytes of ram are to be copied to PROM from ram situated at 1000 hex, then enter the following:

0000	10
0001	00
0002	13
0003	ff

The program and listing are self explanatory. The instruction C (copy from PROM to ram) must be entered twice, this is to prevent accidental overwrite of ram by accidentally pressing C.

Warning

1) To check all voltages before trusting your valuable copy of '2708' softbear to the socket.
2) Check out that the programming pulses are within the 2708 specification, by using a scope while running 'P' of Softbear.

....

For a detailed discussion of 2708 see Personal Computing World Vol. 1 No. 3 P.56.
Putting Bits in their place. D. V. Goadby.

....

U.V. Lamps (bulbs only available from Newbear at £10.00 each.)

ADDRESS	MACHINE CODE	LABEL	OPERATOR & OPERAND	COMMENTS
				This program copies contents of
				PROM device in PROM PROGRAMMER to
				RAM (as stored in START)
				run program from first location
	CE0100		LDX #START	enter RAM address here
	86FF		LDAA #$FF	note:
	B78102		STAA PIABD	modify PIA as required
	C634		LOAB #$34	
	F78101		STAB PIAAC	
	F78103		STAB PIABC	
	7F0000		CLR ADDR	This program to be used when no
	7F0001		CLR ADDR+1	PROM socket is available in
	9600	LOOP	LDAA ADDR	microprocessor system and a copy
	B78102		STAA PIABD	of the PROM Programmer program is
	863C		LDAA #$3C	required to be moved to RAM
	B78101		STAA PIAAC	
	F78101		STAB PIAAC	
	9601		LDAA ADDR+1	
	B78102		STAA PIABD	
	B68100		LDAA PIAAD	
	A700		STAA X	
	08		INX	
	8640		LDAA #$40	(should be 8604)
	7C0001		INC ADDR+1	
	2603		BNE NEXT	
	7C0000		INC ADDR	
	9101	NEXT	CMPA ADDR	(should be 9100)
	26DA		BNE LOOP	
	3F		SWI	

```
          NAM     SOFTBEAR
*********************************************
*
*                    6800
*                  2708 PROM
*                  PROGRAMMER
*               ***************
*
* INSTRUCTIONS:
*
*   USING MIKBUG OR EQUIVALENT, ENTER:
*   START OF COPY ADDRESS   AT  $0000/1
*   END   OF COPY ADDRESS   AT  $0002/3
*
*   ENSURE "PROGRAM" SWITCH IS OFF BEFORE
*   PUTTING PROM IN SOCKET AND DURING READ
*   CYCLES.
*
* COMMANDS:
*
*   P  PROGRAM PROM FROM MEMORY.
*
*   V  VERIFY CONTENTS OF PROM/MEMORY.
*
*   L  LIST CONTENTS OF PROM.
*
*   C  COPY FROM PROM TO MEMORY.
*
*   M  GO TO MONITOR.
*
*   START PROGRAM AT FIRST LOCATION OF PROM
*********************************************
*
*
*   N E W B E A R     C O M P U T I N G
*
*                 S T O R E
*
*   BONE LANE, NEWBURY BERKSHIRE  RG14 5SH
*
*               0635 - 46898
*
*********************************************
```

```
                    *
                    *                   6800
                    *             2708  PROM  PROGRAMMER
                    *
                    *       INTERFACES TO 6820 PIA
                    *    A SIDE OF PIA CONNECTED TO 2708 DATA
                    *    B SIDE OF PIA CONNECTED TO 2708 A0-A7
                    *       ALSO PB0,PB1 ARE CONNECTED TO INPUTS OF
                    *       D FLIPFLOP, OUTPUTS GO TO A8,A9 OF 2708
                    *       FLIPFLOP IS CLOCKED FROM CA2,
                    *       CB2 SWITCHES 26V PROGRAM PULSE (+VE GOING)
                    *       PROGRAM RUNS 100 LOOPS OF 1MS PULSES
                    *
                    *       PIA IS LOCATED AT $ 8100/3
                    *
                    * PROGRAM CAN BE LOCATED ANYWHERE IN MEMORY AS
                    *          IT IS POSITION INDEPENDENT
                    *
                    ***********************************************
                            OPT    O,S,NOG
A048                        ORG    $A048
A048 A100                   FDB    BEGIN    START ADDRESS
0000                        ORG    $0000    RAM AREA
0000 0002           SADD    RMB    2        RAM START OF COPY
0002 0002           EADD    RMB    2        RAM  END  OF COPY
0004 0004           ADDR    RMB    4        SCRATCH AREA
0008 0001           TALLY   RMB    1
     8100           PIAAD   EQU    $8100    A SIDE PIA
     8101           PIAAC   EQU    PIAAD+1
     8102           PIABD   EQU    $8102    B SIDE PIA
     8103           PIABC   EQU    PIABD+1
                    *       MIKBUK/RT68/SWTBUG LINKS
     E07E           OUTMES  EQU    $E07E    OUTPUT STRING UNTIL ($04)
     E1D1           OUTCH   EQU    $E1D1    OUTPUT SINGLE CHARACTER
     E1AC           INCH    EQU    $E1AC    INPUT  SINGLE CHARACTER
     E0C8           OUT4HX  EQU    $E0C8    OUTPUT 4 HEX CHARACTERS
     E0CA           OUT2HX  EQU    $E0CA    OUTPUT 2 HEX CHARACTERS
                    *
                    *
                    *
                    *
                    *
                    *
```

```
A100                    ORG     $A100    START ADDRESS
                   *  PROGRAM CAN BE LOADED ANYWHERE IN MEMORY
A100 8E A047 BEGIN  LDS     #$A047
A103 7F 8101        CLR     PIAAC
A106 7F 8103        CLR     PIABC
A109 86 FF          LDA A   #$FF
A10B B7 8100        STA A   PIAAD
A10E B7 8102        STA A   PIABD
A111 8D 73          BSR     CLR1
A113 7F 0008        CLR     TALLY
A116 C6 0F          LDA B   #M1-M0
A118 8D 70          BSR     MESSG2
A11A CE 0000        LDX     #0
A11D BD E0C8        JSR     OUT4HX
A120 BD E0C8        JSR     OUT4HX
A123 C6 36          LDA B   #M5-M0
A125 8D 63          BSR     MESSG2
A127 BD E1AC        JSR     INCH
A12A 81 4C          CMP A   #'L
A12C 27 5E          BEQ     VER      LIST 2708
A12E 81 50          CMP A   #'P
A130 27 21          BEQ     PROGR    PROGRAM 2708
A132 C6 7F          LDA B   #$7F
A134 D7 08          STA B   TALLY
A136 81 56          CMP A   #'V
A138 27 52          BEQ     VER      VERIFY 2708
A13A 7C 0008        INC     TALLY
A13D 81 43          CMP A   #'C
A13F 26 09          BNE     MM
A141 BD E1AC        JSR     INCH
A144 81 43          CMP A   #'C
A146 27 44          BEQ     VER      COPY
A148 20 B6          BRA     BEGIN
A14A 81 4D   MM     CMP A   #'M
A14C 26 B2          BNE     BEGIN
A14E FE FFFE        LDX     $FFFE
A151 6E 00          JMP     X        RETURN TO MONITOR
A153 DE 00   PROGR  LDX     SADD
A155 DF 04          STX     ADDR
A157 8D 7B          BSR     OUTADD   OUTPUT FIRST
A159 7D 0008        TST     TALLY    DATA & ADDRESS
A15C 26 11          BNE     IN1
A15E C6 80          LDA B   #M2-M0
```

```
01300 A160 8D 28            BSR    MESSG2
01310 A162 BD E1AC          JSR    INCH
01320 A165 81 50            CMP A  #'P
01330 A167 27 02            BEQ    *+4
01340 A169 20 95     BEGIN1 BRA    BEGIN
01350 A16B 86 64            LDA A  #100
01360 A16D 97 08            STA A  TALLY     100 RROGRAM LOOPS
01370 A16F 8D 17     IN1    BSR    PUL1      26V PULSE
01380 A171 8D 61            BSR    OUTADD
01390 A173 09               DEX
01400 A174 9C 02            CPX    EADD      FINISHED?
01410 A176 26 F7            BNE    IN1
01420 A178 8D 0E            BSR    PUL1      PROG LAST LOCATION
01430 A17A 86 2A            LDA A  #'*      OUT ON EVERY LOOP
01440 A17C BD E1D1          JSR    OUTCH
01450 A17F 7A 0008          DEC    TALLY
01460 A182 26 CF            BNE    PROGR
01470 A184 20 E3     IN2    BRA    BEGIN1
01480 A186 20 59     CLR1   BRA    CLEAR     BRANCH LINKS
01490 A188 20 75     PUL1   BRA    PULSE
01500 A18A 20 46     MESSG2 BRA    MESSG3
01510 A18C C6 AF     VER    LDA B  #M3-M0
01520 A18E 8D FA            BSR    MESSG2
01530 A190 7F 8101          CLR    PIAAC
01540 A193 7F 8100          CLR    PIAAD
01550 A196 8D 49            BSR    CLEAR
01560 A198 DE 00            LDX    SADD
01570 A19A 20 07            BRA    NEXT2
01580 A19C DE 04     NEXT   LDX    ADDR
01590 A19E 9C 02            CPX    EADD
01600 A1A0 27 2E            BEQ    END
01610 A1A2 08               INX
01620 A1A3 DF 04     NEXT2  STX    ADDR
01630 A1A5 8D 43            BSR    OUTADR
01640 A1A7 B6 8100          LDA A  PIAAD
01650 A1AA D6 08            LDA B  TALLY
01660 A1AC 27 0A            BEQ    NEXT3     LIST
01670 A1AE 2A 04            BPL    NEXT4     COMP
01680 A1B0 A7 00            STA A  X         COPY
01690 A1B2 20 E8            BRA    NEXT
01700 A1B4 A1 00     NEXT4  CMP A  X
01710 A1B6 27 E4            BEQ    NEXT      INCORRECT MATCH
01720 A1B8 97 06     NEXT3  STA A  ADDR+2
```

```
01730 A1BA A6 00            LDA A  X
01740 A1BC 97 07            STA A  ADDR+3
01750 A1BE C6 BB            LDA B  #M4-M0
01760 A1C0 8D 10            BSR    MESSG3
01770 A1C2 CE 0004          LDX    #ADDR
01780 A1C5 BD E0C8          JSR    OUT4HX
01790 A1C8 BD E0CA          JSR    OUT2HX
01800 A1CB BD E0CA          JSR    OUT2HX
01810 A1CE 20 CC            BRA    NEXT
01820 A1D0 20 B2     END    BRA    IN2      BACK TO BEGIN
01830 A1D2 20 3B     MESSG3 BRA    MESSG
01840                *           OUTPUT ADDRESS & DATA TO 2708
01850 A1D4 DE 04     OUTADD LDX    ADDR
01860 A1D6 A6 00            LDA A  X
01870 A1D8 B7 8100          STA A  PIAAD
01880 A1DB 8D 0D            BSR    OUTADR
01890 A1DD 08        OUT    INX
01900 A1DE DF 04            STX    ADDR
01910 A1E0 39               RTS
01920                *           RESET PIA'S
01930 A1E1 C6 34     CLEAR  LDA B  #$34
01940 A1E3 F7 8101          STA B  PIAAC
01950 A1E6 F7 8103          STA B  PIABC
01960 A1E9 39               RTS
01970 A1EA 8D F5     OUTADR BSR    CLEAR
01980 A1EC 96 04            LDA A  ADDR
01990 A1EE B7 8102          STA A  PIABD
02000 A1F1 86 3C            LDA A  #$3C
02010 A1F3 B7 8101          STA A  PIAAC    PULSE 74175
02020 A1F6 F7 8101          STA B  PIAAC
02030 A1F9 96 05            LDA A  ADDR+1
02040 A1FB B7 8102          STA A  PIABD
02050 A1FE 39               RTS
02060 A1FF 8D E0     PULSE  BSR    CLEAR
02070 A201 86 3C            LDA A  #$3C
02080 A203 B7 8103          STA A  PIABC    UP PROG PULSE
02090 A206 CE 0064          LDX    #100
02100 A209 09        DELAY  DEX             4
02110 A20A 01               NOP             2 = 10 US
02120 A20B 26 FC            BNE    DELAY    4
02130 A20D 20 D2            BRA    CLEAR    DOWN PROG PULSE
02140 A20F 4F        MESSG  CLR A
02150 A210 8D 0D            BSR    ME1
```

```
02160 A212 EB 01    M0     ADD B  1,X
02170 A214 A9 00           ADC A  X
02180 A216 E7 01           STA B  1,X
02190 A218 A7 00           STA A  X
02200 A21A EE 00           LDX    X
02210 A21C 7E E07E         JMP    OUTMESS
02220 A21F 30       ME1    TSX
02230 A220 39              RTS
02240 A221 0D0A     M1     FDB    $D0A      $0D0A
02250 A223 32              FCC    /2708 PROGRAMMER/
02260 A232 0D0A            FDB    $0D0A
02270 A234 53              FCC    /START-END ADDRESS: /
02280 A247 04              FCB    $4        $04
02290 A248 0D0A     M5     FDB    $D0A
02300 A24A 53              FCC    /SWITCH "PROGRAM" OFF!! /
02310 A261 0D0A            FDB    $0D0A
02320 A263 45              FCC    /ENTER M(MIK),P(PRG),V(VERIFY),L(LIST
02330 A288 2C              FCC    /,C(COPY):/
02340 A291 04              FCB    4   $04
02350 A292 0D0A     M2     FDB    $D0A
02360 A294 50              FCC    /PUT PROM IN SOCKET,PRESS "PROGRAM"/
02370 A2B6 0D0A            FDB    $0D0A
02380 A2B8 20              FCC    / ENTER P/
02390 A2C0 04              FCB    4   $04
02400 A2C1 0D0A     M3     FDB    $D0A   $0D0A
02410 A2C3 41              FCC    /ADDR PR ME/
02420 A2CD 0D0A 00  M4 04  FDB    $D0A,$0004    $0D0A, ($000   $0004)
02430 A2D1                 END
SADD   0000
EADD   0002
ADDR   0004
TALLY  0008
PIAAD  8100
PIAAC  8101
PIABD  8102
PIABC  8103
OUTMES E07E
OUTCH  E1D1
INCH   E1AC
OUT4HX E0C8
OUT2HX E0CA
BEGIN  A100
MM     A14A
```

```
PROGR  A153
BEGIN1 A169
IN1    A16F
IN2    A184
CLR1   A186
PUL1   A188
MESSG2 A18A
VER    A18C
NEXT   A19C
NEXT2  A1A3
NEXT4  A1B4
NEXT3  A1B8
END    A1D0
MESSG3 A1D2
OUTADD A1D4
OUT    A1DD
CLEAR  A1E1
OUTADR A1EA
PULSE  A1FF
DELAY  A209
MESSG  A20F
M0     A212
ME1    A21F
M1     A221
M5     A248
M2     A292
M3     A2C1
M4     A2CD

TOTAL ERRORS 00000
```

NewBear Computing Store

BEAR BAG 14

2708 PROM PROGRAMMER

IC1	74LS175
IC2	7406
IC3	2708 (Pre programmed with "6800" programme.)
TR1	BC107
TR2	BFY51
R1	1K ½ watt
R2	1K ½ watt
R3	3K ½ watt
R4	10K ½ watt
R5	10K ½ watt
R6	2.7K ½ watt
R7	10K ½ watt
R8	240 ½ watt
R9	300 ½ watt
R10	100 ½ watt
C1	100 uf +10v
C2	100 uf +30
C3	100 uf +10
C4	0.001 uf
D1	5.6v Zener
D2	5.6v Zener
D3	5.6v Zener
D4	IN4001
D5	IN4001
D6	IN4001
SK1	Low insertion force socket
S1	Single pole toggle switch
T1	18v - 0 - 18v

Edge Connector

Printed Circuit Board

Design Note 35

DRAWING No.
AS 41300



DRAWN	CHECKED	APPROVED
AC		
DATE 17-7-78'	ISSUE	

TITLE
2708 PROGRAMMER

DRAWING No.
AS 41300

8912FT

PARTS LIST

2708 SOCKET
74LS175
7406
BFY51 (TO5 CAN)
BC107 TO3 CAN
SWITCH
RESISTORS x9 1/2 WATT
ZENERS x3
DIODES x3
ELECTROLYTIC CAP x 3
MAINS TRANSFORMER
PCB CONNECTOR

DESIGN NOTE

EDGE CONNECTOR
PIN DESIGNATION

1 0V	23 PA7
2 +5v	24 " 6
3–6 NC	25 " 5
	26 " 4
	27 " 3
7 NOTCH	28 " 2
8 CB2	29 " 1
9–15 NC	30 " 0
	31 PB0
	32 " 1
	33 " 2
	34 " 7
	35 " 6
	36 " 5
	37 " 4
16 CA2	38 " 3
17–22 NC	39 NC
	40 NC
	41 +12v
	42 0V
	43 +5v



2708 PROM PROGRAMMER